

AutoSave Off Book1 - Excel Search (Alt+Q) นพพร โพธิ์ยุทธนาสคิม

File Home Insert Draw Page Layout Formulas Data Review View Developer Help Power Pivot Share Comments

Clipboard Font Alignment Number Styles Cells Editing Analysis Sensitivity

Calibri 11 A A Wrap Text General Conditional Formatting Format as Table Cell Styles Insert Delete Format Sort & Filter Find & Select Analyze Data Sensitivity

A1

1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27 28 29 30 31 32

A B C D E F G H I J K L M N O P Q R S

# 100 เทคนิค

การใช้งาน

# Functions

ที่จะเปลี่ยนชีวิตคุณไปตลอดกาล

Sheet1

Ready



1

# VLOOKUP

ประเภทที่ต้องการให้ค้นหา  
0 หมายถึง ค่าค้นต้องตรงกันเท่านั้น  
1 หมายถึง ค่าค้นใกล้เคียงที่สุด

(lookup\_value, table\_array, col\_index\_num, [range\_lookup])

ค่าที่ต้องการค้นหา

ตารางข้อมูลที่ต้องการค้นหา

ลำดับที่คอลัมน์ที่ต้องการนำค่ามาแสดงผล

**=VLOOKUP(F3,B3:D10,3,0)**

	A	B	C	D	E	F	G
1							
2		ชื่อเมนูอาหาร	หน่วย	ปริมาณแคลอรี		ชื่อเมนูอาหาร	ปริมาณแคลอรี
3		กระเพาะปลา	ชาม	150		กล้วยคลุกมะพร้าว	300
4		กระเพาะปลาตุ๋นน้ำแดง	ชาม	225			
5		กล้วยไข่	ลูก	40			
6		กล้วยคลุกมะพร้าว	ถ้วย	300			
7		กล้วยฉาบ	ชิ้น	29			
8		กล้วยตาก <b>1</b>	ผล <b>2</b>	<b>3</b> 30			
9		กล้วยทอด	ชิ้น	50			
10		กล้วยน้ำว้า	ลูก	36			

# 2

# ROUND

(number, num\_digits)

ตัวเลขที่ต้องการปัดเศษ

ตำแหน่งทศนิยมที่ต้องการ

	A	B	C
1			
2		343.5274	
3			
4		343.5300	
6		343.5270	
8		343.5200	
10		343.5280	
11			
12			

=ROUND(B2,2)

=ROUND(B2,3)

=ROUNDDOWN(B2,2)

=ROUNDUP(B2,3)

# 3

# COUNT

\*นับเฉพาะข้อมูลที่เป็นตัวเลขเท่านั้น!!!

(value1,[value2],...)

ข้อมูลที่ต้องการนับที่ 1

ข้อมูลที่ต้องการนับที่ 2

ข้อมูลที่ต้องการนับที่ n

**=COUNT(B3:F7)**

	A	B	C	D	E	F	G	H	I
1									
2		ลำดับ	ชื่อ	เพศ	อายุ	อาชีพ			
3		1	Bob	ชาย	34	วิศวกร			
4		2	Jeab	หญิง	35	วิศวกร			
5		3	Beer	หญิง	28	พยาบาล			
6		4	Tao	ชาย	25	ทหาร			
7		5	Pang	หญิง	19	นักศึกษา			
8									

ข้อมูลที่เป็นตัวเลข 10





## 6

# COUNTIFS

\*นับข้อมูลตามเงื่อนไข  
(มากกว่า 1 เงื่อนไข)

(criteria\_range1, criteria1, [criteria\_range2, criteria2], [criteria\_range3, ...])

ช่วงข้อมูลตรวจสอบที่ต้องการนับที่ 1

ช่วงข้อมูลตรวจสอบที่ต้องการนับที่ 2

เงื่อนไขที่ต้องการนับที่ 1

เงื่อนไขที่ต้องการนับที่ 2

**=COUNTIFS(D3:D7,"หญิง", E3:E7,"<30")**

	A	B	C	D	E	F	G	H	I
1									
2		ลำดับ	ชื่อ	เพศ	อายุ	อาชีพ			
3		1	Bob	ชาย	34	วิศวกร			
4		2	Jeab	หญิง	35	วิศวกร			
5		3	Beer	หญิง	28	พยาบาล			
6		4	Tao	ชาย	25	ทหาร			
7		5	Pang	หญิง	19	นักศึกษา			
8									

จำนวนเพศหญิงที่อายุน้อยกว่า 30 **2**

## 7

# COUNTBLANK (range)

\*นับจำนวนช่องที่  
ไม่มีข้อมูล

ช่วงเซลล์ข้อมูลที่ต้องการนับ

**=COUNTBLANK(B3:F7)**

	A	B	C	D	E	F	G	H	I
1									
2		ลำดับ	ชื่อ	เพศ	อายุ	อาชีพ			
3		1	Bob	ชาย	34	วิศวกร			
4		2	Jeab	หญิง	35	วิศวกร			
5		3	Beer		28				
6		4	Tao	ชาย	25	ทหาร			
7		5	Pang		19	นักศึกษา			
8									

จำนวนช่องว่าง  
(ช่องที่ไม่มีข้อมูล) 3

## 8

## IF

(logical\_test, [value\_if\_true], [value\_if\_false])

เงื่อนไขและการทดสอบให้ผลลัพธ์  
เป็น True หรือ False

คำตอบหรือเงื่อนไขถัดไปหาก  
ผลลัพธ์เป็น True

คำตอบหรือเงื่อนไขถัดไปหาก  
ผลลัพธ์เป็น False

=IF(C3>=50,"ผ่าน","ไม่ผ่าน")

	A	B	C	D	E
1					
2		<b>ชื่อ</b>	<b>คะแนน</b>	<b>ผลการสอบ</b>	
3		บ๊อบ	45	ไม่ผ่าน	
4		เบียร์	59	ผ่าน	
5		เจียม	80	ผ่าน	
6		แป๋ง	35	ไม่ผ่าน	
7					

\*ตรวจสอบเงื่อนไขคะแนน  
หาก **มากกว่า**  
หรือเท่ากับ 50  
ให้ "ผ่าน"  
ถ้าไม่เข้าเงื่อนไขให้  
"ไม่ผ่าน"

9

# CEILING

(number, significance)

ตัวเลขที่ต้องการปิดเศษ

หน่วยที่ต้องการปิด  
(ปิดขึ้นครั้งละเท่าไร)

**=CEILING(B2,0.25)**

	A	B	C	D	E
1					
2		5.286		5.50	
3					

5.286

5.052



5.75    5.00    **5.25**    **5.50**    5.75

\*ตัวอย่าง  
ปิดขึ้น  ครั้งละ 0.25

# 10

# FLOOR

(number, significance)

ตัวเลขที่ต้องการปิดเศษ

หน่วยที่ต้องการปิด  
(ปิดลงครั้งละเท่าไร)

**=FLOOR(B2,0.25)**

	A	B	C	D	E
1					
2		5.286		5.25	
3					

5.286

5.052



5.75

**5.00**

**5.25**

5.50

5.75

\*ตัวอย่าง  
ปิดลง  ครั้งละ 0.25



## 12

# SUMPRODUCT

\*หาผลรวม ของผลคูณ

(array1, [array2], [array3], ...)

ช่วงข้อมูลที่ต้องการคูณที่ 1

ช่วงข้อมูลที่ต้องการคูณที่ 2

**=SUMPRODUCT(C3:C6,D3:D6)**

	A	B	C	D	E	F
1						
2		<b>ชื่อสินค้า</b>	<b>จำนวน</b>	<b>ราคาสินค้าต่อหน่วย</b>		<b>ค่าใช้จ่ายรวม</b>
3		สบู่	7	10		400
4		ยาสีฟัน	3	25		
5		น้ำยาล้างจาน	6	20		
6		ผงซักฟอก	9	15		
7						

(7x10)+(3x25)+(6x20)+(9x15)

# SUMPRODUCT

\*หาผลรวมของผลคูณ

(array1, [array2], [array3], ...)

## แบบมีเงื่อนไข

ตัวอย่างต้องการผลรวมผลคูณทุกรายการ  
ยกเว้น สินค้า ยาสีฟัน

**=SUMPRODUCT((B3:B6<>"ยาสีฟัน")\*C3:C6,D3:D6)**

	A	B	C	D	E	F
1						
2		ชื่อสินค้า	จำนวน	ราคาสินค้าต่อหน่วย		ค่าใช้จ่ายรวม
3		สบู่	7	10		325
4		<del>ยาสีฟัน</del>	<del>3</del>	<del>25</del>		
5		น้ำยาล้างจาน	6	20		
6		ผงซักฟอก	9	15		
7						
8						

**ไม่รวมบรรทัดนี้**

$(7 \times 10) + (6 \times 20) + (9 \times 15)$



# OR

(logical1, [logical2], ...)

เงื่อนไขตรวจสอบที่ 1

เงื่อนไขตรวจสอบที่ 2

เงื่อนไขตรวจสอบที่ n

\*\*\* หากเงื่อนไข เป็นจริง เพียง ค่าเดียว จะคืนค่า **TRUE** ทันที

\*ใส่สูตรกดปุ่ม



```
=OR(C4:C7=E4:E7)
```

```
=OR(I4:I7=K4:K7)
```

	A	B	C	D	E
1					
2		คุณสมบัติที่มี			
3			นาย ก		คุณผู้หญิง
4		หน้าตาดี	No	=	Yes
5		มีความโรแมนติก	Yes	=	Yes
6		ไม่เจ้าชู้	No	=	Yes
7		พูดจาสุภาพ	No	=	Yes
8					
9			ผลลัพธ์	=	TRUE

	F	G	H	I	J	K
1						
2			คุณสมบัติที่มี			
3				นาย ก		คุณผู้หญิง
4		หน้าตาดี		No	=	Yes
5		มีความโรแมนติก		No	=	Yes
6		ไม่เจ้าชู้		No	=	Yes
7		พูดจาสุภาพ		No	=	Yes
8						
9				ผลลัพธ์	=	FALSE

# 16

# IFNA

(value, value\_if\_na)

ค่าที่ต้องการทดสอบ

ผลลัพธ์ที่ต้องการให้แสดงเมื่อพบ #N/A

ซ่อน Error #N/A ได้  
ประเภทเดียวกัน

=IFNA(VLOOKUP(G3,B3:E7,4,0), "ไม่พบชื่อสินค้า")

A	B	C	D	E	F	G	H
1							
2	Product	ผลผลิต	ของเสีย	%ของเสีย		Product	%ของเสีย
3	A	2000	30	1.50%		H	ไม่พบชื่อสินค้า
4	B	1500	46	3.07%			
5	C			#DIV/0!			
6	D	5000	30	0.60%			
7	E	3500	150	4.29%			
8							

\*\*\*เมื่อค้นหา **Product H**  
ที่ **ไม่มีอยู่ในตาราง** ข้อมูล  
จะปรากฏ **#N/A**

# IFERROR

(value, value\_if\_error)

ค่าที่ต้องการทดสอบ

ผลลัพธ์ที่ต้องการให้แสดงเมื่อพบ error

#VALUE!  
#NAME?  
#N/A  
#REF!  
#NUM!  
#NULL  
#DIV/0!

ซ่อน Error ได้ ทุกประเภท

=IFERROR(VLOOKUP(G3,B3:E7,4,0), "ไม่มีผลผลิต")

	A	B	C	D	E	F	G	H
1								
2		Product	ผลผลิต	ของเสีย	%ของเสีย		Product	%ของเสีย
3		A	2000	30	1.50%		C	ไม่มีผลผลิต
4		B	1500	46	3.07%			
5		C			#DIV/0!			
6		D	5000	30	0.60%			
7		E	3500	150	4.29%			

\*\*\*เมื่อค้นหา **Product C**  
ที่ **มีอยู่ในตาราง** แต่เจอ  
**#DIV/0!**

# INDEX

(array, row\_num, [column\_num])

ขอบเขตตารางอ้างอิง

ตำแหน่งลำดับเลข Row

ตำแหน่งลำดับเลข Column

\*\*\* ใช้ค้นหาหรือ ตำแหน่ง Cell  
ตามพิกัด [Row, Column]

**=INDEX(B3:E6,3,2)**

	A	B	C	D	E	F	G
1		1	2	3	4		
2		Code	Name	เพศ	ส่วนสูง		
3		1111	Bob	ชาย	175		
4		1112	Nina	หญิง	169		
5		1113	David	ชาย	181		David
6		1114	Carol	หญิง	173		
7							

Diagram illustrating the INDEX function. The array B3:E6 is highlighted with a dashed yellow border. The row number 3 and column number 2 are indicated by red arrows pointing to the cell containing 'David'. A vertical stack of blue circles on the right side of the array is numbered 1 to 4, with a red arrow pointing to the 3rd circle. A yellow box labeled 'David' is connected to the 3rd circle and the cell containing 'David' in the array.





# OFFSET

(reference, rows, cols, [height], [width])

**=SUM(OFFSET(A1,2,2,G3,1))**

เซลล์อ้างอิง

เซลล์เริ่มนับตามลำดับแถวอ้างอิงจาก Reference

เซลล์เริ่มนับตามลำดับคอลัมน์อ้างอิงจาก Reference

ขอบเขตความสูงของพื้นที่เซลล์ที่ต้องการ

ขอบเขตความกว้างของพื้นที่เซลล์ที่ต้องการ

ขอบเขตความกว้างของพื้นที่เซลล์ที่ต้องการ

จำนวนรวมสะสมตั้งแต่ Code ที่

1 ถึง 5 = 3,550

ขอบเขตความสูงของพื้นที่เซลล์ที่ต้องการ

$$500+1500+250+400+900 = 3550$$

\*\*\* เราสามารถเลือกช่วงเซลล์ตามต้องการได้โดยเปลี่ยนเลขตรงนี้

	A	B	C	D	E	F	G	H	I
1									
2									
3									
4									
5									
6									
7									
8									
9									

Code	จำนวน
1	500
2	1,500
3	250
4	400
5	900
6	1,200

# INDIRECT

(ref\_text, [a1])

ตำแหน่งที่ต้องการอ้างอิง

ชนิดการอ้างอิง  
ใส่เลข 1 หรือ ไม่ระบุ เป็นแบบ A1  
ใส่เลข 0 เป็นแบบ R1C1

**=INDIRECT("C6")**

	A	B	C	D	E	F
1						
2		<b>ข้อมูลผลผลิต ปี 2018</b>				
3		Month	ผลผลิต (ชิ้น)			ผลผลิต (ชิ้น)
4		Jan	25,000			
5		Feb	20,000			
6		Mar	23,000			23,000
7		Apr	26,000			
8		May	38,000			

# 23

# CHOOSE

(index\_num, value1, [value2], ...)






หมายเลขอ้างอิงค่าหรือสูตรที่ต้องการใช้

```
=CHOOSE(G5,C5,C6,C7,C8,C9)
```

ค่าหรือสูตรที่ต้องการ  
ให้แสดง ลำดับที่ 1

ค่าหรือสูตรที่ต้องการ  
ให้แสดง ลำดับที่ 2

ค่าหรือสูตรที่ต้องการ  
ให้แสดง ลำดับที่ n






	A	B	C	D	E	F	G	H
1								
2								
3								
4			Number	Name	Picture		Number	Name
5			1	Miss A			2	Miss B
6			2	Miss B				
7			3	Miss C				
8			4	Miss D				
9			5	Miss E				

## 24

# CHOOSE+VLOOKUP

\*\*\* ท้ายข้อจำกัดของ VLOOKUP ให้สามารถ LOOKUP ค่าจากขวาไปซ้ายได้

```
=VLOOKUP(I3,CHOOSE({1,2},G3:G7,C3:C7),2,0)
```

Number	Name	Picture	น้ำหนัก	ส่วนสูง	ประเทศ
1	Miss A		53	169	ไทย
2	Miss B		54	168	อเมริกา
3	Miss C		51	164	เวเนซุเอลา
4	Miss D		56	167	ฟิลิปปินส์
5	Miss E		48	163	ญี่ปุ่น

ประเทศ	Name
เวเนซุเอลา	Miss C

Diagram illustrating the VLOOKUP formula using CHOOSE to look up values from right to left. The formula is: `=VLOOKUP(I3,CHOOSE({1,2},G3:G7,C3:C7),2,0)`. The diagram shows a table with columns: Number, Name, Picture, น้ำหนัก, ส่วนสูง, ประเทศ. The formula is applied to cell I3, which contains 'เวเนซุเอลา'. The formula looks up 'เวเนซุเอลา' in the 'ประเทศ' column (column G) and returns the corresponding 'Name' (column C), which is 'Miss C'. A red arrow points from the 'ประเทศ' cell in row 3 to the 'Name' cell in row 3. A blue dashed arrow points from the 'Name' cell in row 3 to the 'Name' cell in the result table.

# SUBTOTAL

(function\_num, value1, [value2], ...)

เซลล์ข้อมูลอ้างอิงที่ต้องการคำนวณ

\*\*\* เราสามารถหาผลรวม Total ได้อย่างง่ายดายด้วยฟังก์ชันนี้ครับ

รหัสฟังก์ชัน ที่ต้องการใช้

รวมเซลล์ที่ Hide ไว้

- 1 - AVERAGE
- 2 - COUNT
- 3 - COUNTA
- 4 - MAX
- 5 - MIN
- 6 - PRODUCT
- 7 - STDEV.S
- 8 - STDEV.P
- 9 - SUM
- 10 - VAR.S
- 11 - VAR.P

ไม่รวมเซลล์ที่ Hide ไว้

- 101 - AVERAGE
- 102 - COUNT
- 103 - COUNTA
- 104 - MAX
- 105 - MIN
- 106 - PRODUCT
- 107 - STDEV.S
- 108 - STDEV.P
- 109 - SUM
- 110 - VAR.S
- 111 - VAR.P

	A	B	C
1			
2		บริษัท	จำนวน
3		A	2,000
4		B	5,500
5		C	3,000
6		Total ABC	10,500
7		E	1,000
8		F	2,500
9		G	4,000
10		Total EFG	7,500
11		Sum Total (ABCEFG)	18,000
12			

=SUBTOTAL(9,C3:C5)

=SUBTOTAL(9,C7:C9)

=SUBTOTAL(9,C3:C10)



# 27

# SUMARRAY

## หลักการทำงาน

$$=SUM(\{7,3,6,9\}*\{10,25,20,15\})$$

ลำดับที่ 4  
ลำดับที่ 3  
ลำดับที่ 2  
ลำดับที่ 1


ลำดับที่ 4  
ลำดับที่ 3  
ลำดับที่ 2  
ลำดับที่ 1

1 2 3 4

$$= (7 \times 10) + (3 \times 25) + (6 \times 20) + (9 \times 15)$$

$$= 70 + 75 + 120 + 135$$

$$= 400$$




$$=SUM(C3:C6*D3:D6)$$

	A	B	C	D	E	F
1						
2		<b>ชื่อสินค้า</b>	<b>จำนวน</b>	<b>ราคาสินค้าต่อหน่วย</b>		<b>ค่าใช้จ่ายรวม</b>
3		สบู่	7	10		400
4		ยาสีฟัน	3	25		
5		น้ำยาล้างจาน	6	20		
6		ผงซักฟอก	9	15		
7						

## 28

## SUMIFARRAY

\*\*\* หาผลรวมแบบมีเงื่อนไขโดยใช้  
หลักการ Array Formula




**=SUM(IF((D3:F3=H3),D4:F7\*C4:C7))**

	A	B	C	D	E	F	G	H	I
1									
2				จำนวนที่ซื้อของแต่ละคน			ชื่อผู้ซื้อ	ค่าใช้จ่ายรวม	
3		ชื่อสินค้า	ราคาสินค้าต่อหน่วย	Bob	Jeab	Beer	Jeab	245	
		สบู่	10	7	8	4			
		ยาสีฟัน	25	3	2	6			
		น้ำยาล้างจาน	20	6	2	5			
		งชั๊กฟอก	15	9	5	9			

$$=(10 \times 8) + (25 \times 2) + (20 \times 2) + (15 \times 5) = 245$$



# MAXIFARRAY



**=MAX(IF(C5:C20=F5,D5:D20))**

\*\*\* หากค่าสูงสุดแบบมีเงื่อนไข  
โดยใช้หลักการ Array Formula


	A	B	C	D	E	F	G
1							
2							
3							
4		Date	Product	Total Revenue		Product	Max Revenue
5		Sep-17	โพสอิท	33,250		สมุดโน้ต	4,480
6		Nov-17	ไฮไลท์	4,000			
7		Dec-17	สมุดโน้ต	2,880			
8		Mar-18	ไฮไลท์	2,700			
9		Apr-18	สมุดโน้ต	4,480			
10		Jun-18	ดินสอ	21,000			
11		Aug-18	ไฮไลท์	3,750			
12		Oct-18	แฟ้ม	7,200			
13		Oct-18	โพสอิท	3,200			
14		Nov-18	แฟ้ม	21,000			
15		Feb-19	ไฮไลท์	7,200			
16		May-19	โพสอิท	10,150			
17		Jul-19	ดินสอ	28,350			
18		Aug-19	แฟ้ม	2,800			
19		Oct-19	ดินสอ	8,700			
20		Nov-19	สมุดโน้ต	800			
21							

หา ยอดขาย  
ที่สูงที่สุด ของ  
สินค้า  
"สมุดโน้ต"




# MULTIPLE VLOOKUP

Labor Energy Welfare Maintenance



```
=IFERROR(INDEX($D$6:$D$15,  
SMALL(IF(F$5=$C$6:$C$15,  
ROW($B$6:$B$15)-5),  
ROW()-5)), "")
```

Expense Type	Department
Labor	Production
Energy	Engineering
Welfare	Production
Labor	QA
Maintenance	Engineering
Welfare	QA
Labor	QC
Welfare	HR
Energy	Production
Welfare	QC




A B C D E F G H I

	A	B	C	D	E	F	G	H	I
1									
2									
3									
4									
5									
6									
7									
8									
9									
10									
11									
12									
13									
14									
15									
16									

Expense Type	Department
Labor	Production
Energy	Engineering
Welfare	Production
Labor	QA
Maintenance	Engineering
Welfare	QA
Labor	QC
Welfare	HR
Energy	Production
Welfare	QC



Labor	Energy	Welfare	Maintenance
Production	Engineering	Production	Engineering
QA	Production	QA	
QC		HR	
		QC	



# ROW

([reference])

ใช้หา ตัวเลขลำดับแถว  
ของเซลล์ที่อ้างอิง

เซลล์อ้างอิง (ไม่ใส่อะไรเลยก็ได้)

-2 หมายถึง 2 คอลัมน์  
ตรงนี้ครับ

**=ROW( )-2**

	A	B	C	D	E
1					
2					
3			<b>No.</b>	<b>Product</b>	<b>Value</b>
4			1	AAA	103.98
5			2	BBB	212.74
6			3	CCC	75.74
7			4	GGG	89.76
8			5	HHH	121.76
9			6	III	353.77
10			7	JJJ	282.76

=ROW( )-2

=ROW( )-2

.

.

=ROW( )-2



# ROWS (array)

ใช้ หาจำนวนแถว  
ของช่วงเซลล์ ที่อ้างอิง

ช่วงเซลล์อ้างอิง (บังคับต้องใส่)

สูตรแต่ละบรรทัดก็จะเป็น  
แบบนี้เลยครึบ



เซลล์ B3 จะถูก Lock เอาไว้

=ROWS(B\$3:B3)

=ROWS(B\$3:B4)

=ROWS(B\$3:B5)

=ROWS(B\$3:B6)

=ROWS(B\$3:B7)

=ROWS(B\$3:B8)

=ROWS(B\$3:B9)

=ROWS(B\$3:B10)

=ROWS(B\$3:B11)

=ROWS(B\$3:B12)

## =ROWS(B\$3:B3)

	A	B	C	D	E	F
1						
2						
3						
4						
5						
6						
7						
8						
9						
10						
11						
12						
13						

No.	Product	Value
1	AAA	103.98
2	BBB	212.74
3	CCC	75.74
4	DDD	91.78
5	EEE	324.48
6	FFF	257.76
7	GGG	89.76
8	HHH	121.76
9	III	353.77
10	JJJ	282.76


แล้วเราก็อาศัย  
**หลักการนับ**  
**จำนวนช่องเซลล์**  
**ของ ROWS** มา  
**Run Number** ให้  
เรา นั่นเองครึบ





# CHAR(10)

**=B3&CHAR(10)&C3&CHAR(10)&D3**



F6    X ✓ f    =B3&CHAR(10)&C3&CHAR(10)&D3

A    B    C    D    E    F    G

1

2    **ข้อความที่ 1**    **ข้อความที่ 2**    **ข้อความที่ 3**

3    ผม    รัก    คุณ

4

5

6

7

รวมแล้วตัด  
ข้อความ  
ให้อยู่คนละ  
บรรทัด

**รวมข้อความ**

ผม

รัก

คุณ



36

# YEARFRAC

(start\_date, end\_date, [basis])

วันที่เริ่มต้น

วันที่สิ้นสุด

ชนิดของหลักเกณฑ์ในการนับจำนวนวันที่ใช้

**=INT(YEARFRAC(E5,TODAY()))**

## รายชื่อพนักงานบริษัทวิศวกรสร้างภาพ

No.	Name	Surname	Birthday	Current Age
1	Wanchai	Meena	15-May-1982	37
2	Veerayut	Sudsomwang	12-Jul-1999	20
3	Chanida	Jantee	09-Aug-1988	31
4	Patcharin	Thayala	20-Oct-2000	19
5	Udomsuk	Samutsakorn	02-Feb-1975	44



# 38

# REPT

(text, number\_times)

คำหรือตัวอักษรที่ต้องการพิมพ์ซ้ำ

จำนวนคำที่ต้องการพิมพ์ซ้ำ




สั่งพิมพ์คำหรือตัวอักษรซ้ำๆ  
ได้ตามจำนวนที่ต้องการ

```
D5=REPT("|",C5)
```

คะแนนสอบวิชา Excel

ชื่อ-สกุล	คะแนน	
นพพร นอนมา	75	
กฤษฎา พารวย	30	
นฤมล คนสวย	56	
สำรวย คนงาม	100	
เนตรสุดา น่ารัก	39	
ประจักษ์ เกรงขาม	90	
นพดล ติดตาม	25	
สงคราม ย่ามใจ	64	

เปลี่ยนตัวอักษรเป็น  
Vonique 64







41

# SUMPRODUCT

(array1, [array2], [array3], ...)

**=SUMPRODUCT(--(C5:C14=F5))**

ช่วงข้อมูลที่ต้องการคูณที่ 1

ช่วงข้อมูลที่ต้องการคูณที่ 2

ใช้ **Sumproduct** แทน  
**Countif**

นับแทน Count ได้  
จริงๆ ด้วยอะ:

นับแทน เป็น  
แผนได้มัย อีอิ



		=SUMPRODUCT(--(C5:C14=F5))					
	A	B	C	D	E	F	G
1							
2		<b>บันทึกการขาย</b>					
3							
4		<b>No.</b>	<b>สินค้า</b>			<b>สินค้า</b>	<b>จำนวนรายการ</b>
5		1	แก้อี้			แฟ้ม	4
6		2	โต๊ะ				
7		3	แฟ้ม				
8		4	โต๊ะ				
9		5	แฟ้ม				
10		6	แก้อี้				
11		7	แฟ้ม				
12		8	แฟ้ม				
13		9	โต๊ะ				
14		10	แก้อี้				
15							

# AVERAGEIF

\*คำนวณค่าเฉลี่ย ไม่รวมค่า Error

**=AVERAGEIF(C3:C10,">0")**

หากไม่ใส่ค่าใดๆ ที่อาร์กิวเมนต์นี้ ฟังก์ชันจะ  
ใช้ค่า Range เป็นค่าคำนวณแทน

(range, criteria, [average\_range])

ช่วงข้อมูลเงื่อนไข

เงื่อนไข

ช่วงข้อมูลที่ต้องการคำนวณ

	A	B	C	D	E
1					
2		Day	Temperature		Average Temp.
3		1	36.5	✓	34.4
4		2	35.3	✓	
5		3	33.4	✓	
6		4	#N/A	✗	
7		5	#DIV/0!	✗	
8		6	32.3	✓	
9		7	#N/A	✗	
10		8	34.6	✓	
11					

เครื่องหมาย **มากกว่า**  
หรือ **น้อยกว่า** ปรับตาม  
ข้อมูลตัวเลข **บวกหรือ**  
**ลบ** ได้เลยนะครับ



43

# VLOOKUP ARRAY



```
=VLOOKUP(I3,B3:F11,{2,3,4,5},0)
```

ID	Name	Surname	Department	Salary
A0001	Somchai	Saisamorn	Production	32,429
A0002	Khunchorn	Gnodgnam	HR	17,203
A0003	Toomtarn	Meenoi	Sale	29,860
A0004	Kloy	Raktinderm	Marketing	26,538
A0005	Boonterm	Permsatja	QA	18,690
A0006	Jintana	Lasamlet	QC	34,596
A0007	Surashet	Yeamsooksan	Warehouse	34,662
A0008	Koonlanan	Janta	Purchasing	23,031
A0009	Sartra	Pajareuan	Accounting	31,162

ID
A0007

Name	Surname	Department	Salary
Surashet	Yeamsooksan	Warehouse	34,662

44

# LEFT+FIND

บุญเต็ม เพิ่มสัจจา

\*แยกชื่อ ออกจากนามสกุล

**=LEFT(B3,FIND(" ",B3)-1)**

	A	B	C	D
1				
2		ชื่อ นามสกุล		ชื่อ
3		บุญเต็ม เพิ่มสัจจา		บุญเต็ม
4		จินตนา ลาส้า		จินตนา
5		สุรเชษฐ์ เยี่ยมสุขสันต์		สุรเชษฐ์
6		กุลนันท์ จันทรา		กุลนันท์
7		ศาสตรา พาเจริญ		ศาสตรา
8				

FIND  
LEFT RIGHT LEN



# RIGHT+LEN+FIND

\*แยกนามสกุล ออกจากชื่อ



=RIGHT(B3,LEN(B3)-FIND(" ",B3))

	A	B	C	D	E
1					
2		<b>ชื่อ นามสกุล</b>			
3		บุญเต็ม เพิ่มสัจจา		<b>ชื่อ</b>	<b>นามสกุล</b>
4		จินตนา ลาส้า		บุญเต็ม	เพิ่มสัจจา
5		สุรเชษฐ์ เยี่ยมสุขสันต์		จินตนา	ลาส้า
6		กุลนันท์ จันทรา		สุรเชษฐ์	เยี่ยมสุขสันต์
7		ศาสตรา พาเจริญ		กุลนันท์	จันทรา
8				ศาสตรา	พาเจริญ

46

# Distinct Count

ARRAY

```
=SUM(1/COUNTIF(C3:C10,C3:C10))
```


	A	B	C	D	E	F
1						
2		วันที่	สินค้า	จำนวน		Distinct Count
3		07/08/20	น้ำอัดลม	5		4
4		08/08/20	น้ำอัดลม	10		
5		09/08/20	ปลากระป๋อง	2		
6		10/08/20	ปลากระป๋อง	2		
7		11/08/20	นมสด	7		
8		12/08/20	บะหมี่กึ่งสำเร็จรูป	25		
9		13/08/20	น้ำอัดลม	8		
10		14/08/20	บะหมี่กึ่งสำเร็จรูป	14		

Ctrl + Shift + Enter

นับข้อมูลแบบไม่ซ้ำ

# ROUND

\*ปิดเศษหลักที่ ไม่ใช่  
ทศนิยม



	A	B	C
1			
2		ค่าที่ต้องการปิด	คำตอบที่ต้องการ
3		8,796	9,000
4		12,211	12,000
5		345,035	345,000
6			



# RANK

(number, ref, [order])

ตัวเลขที่ต้องการจัดอันดับ

กลุ่มตัวเลขอ้างอิง

ประเภทการจัดเรียง

0 - Descending (เรียงจากมากไปน้อย)

1 - Ascending (เรียงจากน้อยไปมาก)

**=RANK(C3,\$C\$3:\$C\$10)**

	A	B	C	D
1				
2		<b>ชื่อ-นามสกุล</b>	<b>คะแนนสอบ</b>	<b>อันดับที่</b>
3		ด.ช.วิชัย มากมี	55	5
4		ด.ญ.ภาวิณี ศรีสม	74	3
5		ด.ญ.สุดารัตน์ พัดลม	82	2
6		ด.ช.พรพรหม ยมดี	55	5
7		ด.ช.อุเทน จันฉาย	90	1
8		ด.ช.สมหมาย ระบายสี	45	7
9		ด.ญ.จงรัก พักดี	32	8
10		ด.ญ.สมศรี สบายใจ	57	4
11				



49

# PMT

(rate, nper, pv, [fv], [type])

อัตราดอกเบี้ย

ระยะเวลาผ่อนชำระ

เงินต้น

กำหนดชำระเงิน

ต้นงวด/ปลายงวด [1,0]

มูลค่าในอนาคตหรือ  
เงินสดคงเหลือ



หมายถึง 12 เดือน

**=PMT(B3/12,C3\*12,D3)**

	A	B	C	D	E	F
1						
2		อัตราดอกเบี้ยต่อปี	ระยะเวลาผ่อนชำระ (ปี)	เงินต้น	อัตราผ่อนชำระต่อเดือน	รวมเงินที่ชำระทั้งหมด
3		6.5%	2	100,000	-B 4,454.63	฿ 106,911.00
4		7.2%	4	100,000	-B 2,403.91	฿ 115,387.89
5		9.9%	8	100,000	-B 1,512.13	฿ 145,164.50
6						

\*คำนวณ อัตราผ่อนชำระ ต่อเดือน



50

# INDEX+MATCH

## ARRAY



=INDEX(D3:D9,MATCH(H2&H3,B3:B9&C3:C9,0))

	A	B	C	D	E	F	G	H
1								
2		<b>First Name</b>	<b>Last Name</b>	<b>Salary</b>		<b>First Name</b>	Andrews	
3		Jones	Lee	฿ 45,000		<b>Last Name</b>	Smith	
4		Kivell	Lewis	฿ 68,000		<b>Salary</b>	฿ 95,000	
5		Jones	Walker	฿ 50,000				
6		Gill	Lopez	฿ 89,000				
7		Gill	Lapota	฿ 53,000				
8		Andrews	Smith	฿ 95,000				
9		Kivell	Anserson	฿ 40,000				
10								
11								

The image shows a spreadsheet with two tables. The left table (A2:D9) lists employees with columns for First Name, Last Name, and Salary. The right table (G2:H3) shows a search for 'Andrews Smith' and returns the salary '฿ 95,000'. A yellow arrow points from the search criteria in G2:H3 to the corresponding row in the left table (A8:D8). A yellow arrow also points from the salary value in H3 to the corresponding cell in the left table (D8).

51

# ROUNDUP

คำนวณหาเลข Quarter  
ง่ายๆ

= "Q" & ROUNDUP(MONTH(B3)/3,0)

	A	B	C
1			
2		<b>Date</b>	<b>Quarter</b>
3		10/03/2020	Q1
4		08/05/2020	Q2
5		26/05/2020	Q2
6		12/07/2020	Q3
7		23/09/2020	Q3
8		02/11/2020	Q4
9		21/11/2020	Q4
10			



52


# SUMIFS+INDEX+MATCH

ดึงค่าในคอลัมน์ที่ต้องการตามเงื่อนไข ง่ายๆ

```
=SUMIFS(INDEX($D$6:$S$13,0,
MATCH(C$16,$D$5:$S$5,0))
,$C$6:$C$13,C$17,$B$6:$B$13,$B18)
```

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S
3																			
4				Year 2020															
5	Product	T/A	Jan	Feb	Mar	Q1	Apr	May	Jun	Q2	Jul	Aug	Sep	Q3	Oct	Nov	Dec	Q4	
6	A0001	Target	500	500	500	1500	500	500	500	1500	500	500	500	1500	500	500	500	1500	
7		Actual	247	254	450	951	360	462	429	1251	234	475	439	1148	275	356	325	956	
8	A0002	Target	300	300	300	900	300	300	300	900	300	300	300	900	300	300	300	900	
9		Actual	495	493	471	1459	431	321	383	1135	448	296	235	979	339	334	411	1084	
10	B0001	Target	400	400	400	1200	400	400	400	1200	400	400	400	1200	400	400	400	1200	
11		Actual	275	405	233	913	249	313	386	948	301	208	385	894	343	308	397	1048	
12	B0002	Target	500	500	500	1500	500	500	500	1500	500	500	500	1500	500	500	500	1500	
13		Actual	360	411	230	1001	495	440	380	1315	215	362	460	1037	277	208	220	705	

	Q1	Q1	Q2	Q2	Q3	Q3	Q4	Q4
Product	Target	Actual	Target	Actual	Target	Actual	Target	Actual
A0001	1500	951	1500	1251	1500	1148	1500	956
A0002	900	1459	900	1135	900	979	900	1084
B0001	1200	913	1200	948	1200	894	1200	1048
B0002	1500	1001	1500	1315	1500	1037	1500	705





# AND + Conditional Formatting



**=AND(E\$5>=\$C6,E\$5<=\$D6)**

สร้าง

**Automatic  
Gantt Chart**

The screenshot shows an Excel spreadsheet with the following data:

ขั้นตอน	เริ่ม	สิ้นสุด	1	2	3	4	5	6
ชื่อโครงการ	01-Jan-20	05-Jan-20						
หลักการและเหตุผล	06-Jan-20	09-Jan-20						
เป้าหมาย	09-Jan-20	12-Jan-20						
การดำเนินงาน	12-Jan-20	14-Jan-20						
ติดตามผล	15-Jan-20	16-Jan-20						

The 'New Formatting Rule' dialog box is open, showing the formula `=AND(E$5>=$C6,E$5<=$D6)` in the 'Format values where this formula is true:' field.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T	U	V
1																						
2		แผนการดำเนินการโครงการ																				
3					Jan-20																	
4		ขั้นตอน	เริ่ม	สิ้นสุด	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18
5		ชื่อโครงการ	01-Jan-20	05-Jan-20	█	█	█	█	█													
6		หลักการและเหตุผล	06-Jan-20	09-Jan-20						█	█	█	█									
7		เป้าหมาย	09-Jan-20	12-Jan-20									█	█	█	█						
8		การดำเนินงาน	12-Jan-20	14-Jan-20												█	█	█				
9		ติดตามผล	15-Jan-20	16-Jan-20															█	█		



## เชื่อมคำและปรับ Format ให้ถูกต้อง

= "ขาย"&C3&" วันที่ "&TEXT(B3,"dd/mm/yy")&" เป็นเงิน "  
&TEXT(D3,"#,#" )&" บาท"

	A	B	C	D	E	F
1						
2		<b>Date</b>	<b>สินค้า</b>	<b>ยอดขาย</b>		<b>เชื่อมคำสร้างประโยค</b>
3		20/12/2020	ผงซักฟอก	10,000		ขายผงซักฟอก วันที่ 20/12/20 เป็นเงิน 10,000 บาท
4		21/12/2020	ยาสีฟัน	5,000		ขายยาสีฟัน วันที่ 21/12/20 เป็นเงิน 5,000 บาท
5		22/12/2020	น้ำยาล้างจาน	7,500		ขายน้ำยาล้างจาน วันที่ 22/12/20 เป็นเงิน 7,500 บาท
6						
7						



## 56

# COUNTIFS

แบ่งกลุ่มและนับจำนวนตามช่วงข้อมูลง่ายๆ

```
=COUNTIFS($C$5:$C$104,">="&H5
,$C$5:$C$104,"<="&I5)
```

	A	B	C	D	E	F	G	H	I	J	K	
1												
2	<b>ข้อมูลอายุของพนักงานจำนวน 100 คน</b>											
3												
4		<b>คนที่</b>	<b>อายุ</b>					<b>กลุ่มที่</b>	<b>ช่วงอายุ</b>	<b>เริ่มต้น</b>	<b>สิ้นสุด</b>	<b>จำนวน</b>
5		1	43					1	21 - 25	21	25	2
6		2	41					2	26 - 30	26	30	9
7		3	30					3	31 - 35	31	35	19
8		4	34					4	36 - 40	36	40	20
9		5	57					5	41 - 45	41	45	17
10		6	47					6	46 - 50	46	50	14
11		7	48					7	51 - 55	51	55	11
12		8	37					8	56 - 60	56	60	8
13		9	33									
14		10	39									

# VLOOKUP

หาค่าสุดท้าย เมื่อกรอกข้อมูลใหม่แบบอัตโนมัติ

**=VLOOKUP(9.99999999999999999E+307,C:C,TRUE)**

	A	B	C	D	E	F
1						
2		คะแนนสอบรวม				
3						
4		ชื่อ	คะแนน		คะแนน	
5		บุญเต็ม	64		99	
6		จินตนา	51			
7		สรเชษฐ์	45			
8		กุลนันท์	78			
9		ศาสตรา	58			
10		ศราวุธ	65			
11		องอาจ	84			
12		ชินวดี	92			
13		ธิดารัตน์	78			
14		นภาพร	99			
15						
16						
17						

58

# VLOOKUP + Wildcard

🔍 ค้นหาได้แม้คีย์คำค้นแค่บางส่วน



**=VLOOKUP("\*"&D3&"\*",B3:B7,1,FALSE)**

	A	B	C	D	E
1					
2		<b>ชื่อ-นามสกุล</b>		<b>คำค้น</b>	<b>ผลลัพธ์</b>
3		บุญเต็ม เพิ่มสัจจา		สุรเช	สุรเชษฐ์ เยี่ยมสุขสันต์
4		จินตนา ลาส้า			
5		สุรเชษฐ์ เยี่ยมสุขสันต์			
6		กุลนันท์ จันทรา			
7		ศาสตรา พาเจริญ			
8					
9					

# HLOOKUP + MATCH

 ค้นหาแนวนอนอย่างง่ายดาย

```
=HLOOKUP(C$10,$C$2:$N$5,  
MATCH($B11,$B$2:$B$5,0),FALSE)
```

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1															
2		<b>Type/Month</b>	<b>Jan-19</b>	<b>Feb-19</b>	<b>Mar-19</b>	<b>Apr-19</b>	<b>May-19</b>	<b>Jun-19</b>	<b>Jul-19</b>	<b>Aug-19</b>	<b>Sep-19</b>	<b>Oct-19</b>	<b>Nov-19</b>	<b>Dec-19</b>	<b>Total</b>
3		<b>Units Sold</b>	106	238	26	102	221	74	42	21	90	41	13	263	1,237
4		<b>Expense</b>	16,100	31,850	3,200	14,880	32,240	10,350	5,850	3,150	11,960	4,950	1,650	35,700	171,880
5		<b>Cost Per Unit</b>	152	134	123	146	146	140	139	150	133	121	127	136	139
6															
7															
8		<b>2 Months Compare</b>													
9															
10		<b>Type/Month</b>	<b>Jun-19</b>	<b>Dec-19</b>											
11		<b>Units Sold</b>	74	263											
12		<b>Expense</b>	10,350	35,700											
13		<b>Cost Per Unit</b>	140	136											
14															
15															





# LEN (text)

1 2 3 4 5

ข้อความที่ต้องการนับ

=LEN(B3)

นับจำนวนตัวอักษรและวรรณยุกต์ทั้งหมด ในข้อความ  
(ช่องว่าง 1 เคาะ ก็นับเป็น 1 ตัวอักษรด้วย)

	A	B	C
1			
2		Name Surname	จำนวนตัวอักษรและ วรรณยุกต์ทั้งหมด
3		Somchai Saisamorn	17
4		Khunchorn Gnodgnam	18
5		Toomtarn Meenoi	14
6		Kloy Raktinderm	15
7		บณเดิม เพิ่มสัจจา	18
8		จินตนา ลาสำเร็จ	15
9		สรเชษฐ์ เขียมสุขสันต์	23
10		กุลนันท์ จันทรา	15
11		ศาสตรา พาเจริญ	14
12			

8

1 2 3 4 5 6 7 9 10 11 12 13 14 15 16 17

Somchai Saisamorn

สระอำ = 1 ตัวอักษร


7

1 2 3 4 5 6 8 9 10 11 12 13 14 15

จินตนา ลาสำเร็จ

# 62

# FIND



ค้นหาตำแหน่งของคำหรือ  
ตัวอักษรที่ต้องการ

(และมีเงื่อนไขว่าตัวอักษรพิมพ์เล็ก  
พิมพ์ใหญ่ต้องตรงกันเท่านั้น)

**A=A**

**b=b**


**A≠a**

(find\_text, within\_text, [start\_num])

คำที่ต้องการค้นหา

ข้อความที่จะค้นหา

เริ่มค้นจากตัวอักษรที่...



**=FIND(D5,B3,8)**

**=FIND(D3,B3,1)**

	A	B	C	D	E
1					
2		ข้อความที่จะค้นหา		คำที่ต้องการค้นหา	ผลลัพธ์
3		Somchai Saisamorn		ai	6
4				ai	10
5				Ai	#VALUE!
6					

เริ่มค้นคำว่า "ai" จากอักษรที่ 1 จะได้คำตอบ 6  
 เริ่มค้นคำว่า "ai" จากอักษรที่ 8 จะได้คำตอบ 10  
 เริ่มค้นคำว่า "Ai" จากอักษรที่ 8 จะได้คำตอบ #VALUE!  
 เพราะค้นหาไม่เจอ เนื่องจากตัว A เป็นอักษรพิมพ์ใหญ่

# SEARCH

(find\_text, within\_text, [start\_num])



ค้นหาตำแหน่งของคำหรือ  
ตัวอักษรที่ต้องการ  
(ไม่มีเงื่อนไขเรื่องอักษร  
พิมพ์เล็กพิมพ์ใหญ่)

A=A

b=b

A=a

คำที่ต้องการค้นหา

ข้อความที่จะค้นหา

เริ่มค้นจากตัวอักษรที่...

1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17

Somchai Saisamorn

=SEARCH(D5,B3,8)

=SEARCH(D3,B3,1)

	A	B	C	D	E
1					
2		<b>ข้อความที่จะค้นหา</b>		<b>คำที่ต้องการค้นหา</b>	<b>ผลลัพธ์</b>
3		Somchai Saisamorn		ai	6
4				ai	10
5				Ai	10
6					

เริ่มค้นคำว่า "ai" จากอักษรที่ 1 จะได้คำตอบ 6  
เริ่มค้นคำว่า "ai" จากอักษรที่ 8 จะได้คำตอบ 10  
เริ่มค้นคำว่า "Ai" จากอักษรที่ 8 จะได้คำตอบ 10  
แม้ว่าจะเป็นตัวอักษรพิมพ์ใหญ่ก็สามารถค้นหาได้

# 64



# TRIM



(text)

ตัดช่องว่าง (ที่อาจมองไม่เห็น) ในข้อความทิ้งไป

- หัวกับท้าย ตัดออกให้หมด
- ส่วน ตรงกลาง ตัดให้เหลือ 1 เคาะ



**=TRIM(B3)**

	A	B	C	D	E	F
1						
2		ข้อความก่อน TRIM	ใช้ฟังก์ชัน LEN		ข้อความหลัง TRIM	ใช้ฟังก์ชัน LEN
3		Mr. Somchai Saisamorn	=LEN(B3)	30	Mr. Somchai Saisamorn	=LEN(E3)
4		Ms. Khunchorn Gnodgnam		25	Ms. Khunchorn Gnodgnam	22
5		Mr. Toomtarn Meenoi		23	Mr. Toomtarn Meenoi	18
6						

65

# CLEAN

(text)



ลบสัญลักษณ์แปลกๆ ออกจากข้อความ  
(แต่ลบไม่ได้ทั้งหมดนะจ๊ะ)

\*\*\* ลบอักขระที่พิมพ์ไม่ได้ 32 ตัวแรกในโค้ด ASCII  
แบบ 7 บิต (ค่า 0 ถึง 31) ออกจากข้อความ

=CLEAN(B3)

	A	B	C	D
1				
2		ข้อความก่อน CLEAN		ข้อความหลัง CLEAN
3		Mr. □ Somchai □ □		Mr.Somchai
4		Ms. □ □ Khunchorn		Ms.Khunchorn
5		Mr.Toomtam □ □ □ □		Mr.Toomtam
6				


# 66

# LEFT

(text, [num\_chars])

ข้อความที่ต้องการตัด

จำนวนตัวอักษรที่ต้องการ โดยเริ่มนับจากด้านซ้าย (ช่องว่าง 1 เคาะนับเป็น 1 ตัวอักษร)



ตัดตัวอักษรด้าน **ซ้าย** ของ  
ข้อความตามจำนวนที่ต้องการ

1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21

**Mr. Somchai Saisamorn**

**=LEFT(B3,11)**

**=LEFT(B3,7)**

**=LEFT(B3,3)**

	A	B	C	D
1				
2		ข้อความ		ข้อความหลังใช้ LEFT
3		Mr. Somchai Saisamorn	Mr.	
4			Mr. Som	
5			Mr. Somchai	
6				


67

# RIGHT


(text, [num\_chars])

ข้อความที่ต้องการตัด

จำนวนตัวอักษรที่ต้องการ โดยเริ่มนับจากด้านขวา  
(ช่องว่าง 1 เคาะนับเป็น 1 ตัวอักษร)



ตัดตัวอักษรด้าน **ขวา** ของ  
ข้อความตามจำนวนที่ต้องการ



**=RIGHT(B3,17)**

**=RIGHT(B3,9)**


**=RIGHT(B3,4)**

	A	B	C	D
1				
2		ข้อความ		ข้อความหลังใช้ RIGHT
3		Mr. Somchai Saisamorn		morn
4				Saisamorn
5				Somchai Saisamorn

68

# MID


(text, start\_num, num\_chars)



ข้อความที่ต้องการตัด

ลำดับของตัวอักษรตัวแรกที่ต้องการนับจากด้านซ้าย

จำนวนตัวอักษรที่ต้องการ โดย เริ่มนับ 1 จาก start\_num (ช่องว่าง 1 เคาะนับเป็น 1 ตัวอักษร)



Mr. Somchai Saisamorn

=MID(B3,13,9)

=MID(B3,5,7)


=MID(B3,5,3)

	A	B	C	D
1				
2		ข้อความ		ข้อความหลังใช้ MID
3		Mr. Somchai Saisamorn		Som
4				Somchai
5				Saisamorn
6				

69

# LOWER

(text)



ปรับตัวอักษร (ภาษาอังกฤษ)  
ให้เป็น ตัวพิมพ์เล็ก ทั้งหมด


=LOWER(B3)

	A	B	C	D
1				
2		Name Surname		หลังใช้ LOWER
3		Somchai Saisamorn		somchai saisamorn
4		Khunchorn Gnodgnam		khunchorn gnodgnam
5		Toomtarn Meenoi		toomtarn meenoi
6		Kloy Raktinderm		kloy raktinderm
7				

# 70

# UPPER

(text)



ปรับตัวอักษร (ภาษาอังกฤษ)  
ให้เป็น ตัวพิมพ์ใหญ่ ทั้งหมด

**=UPPER(B3)**

	A	B	C	D
1				
2		<b>Name Surname</b>		<b>หลังใช้ UPPER</b>
3		Somchai Saisamorn		SOMCHAI SAISAMORN
4		Khunchorn Gnodgnam		KHUNCHORN GNODGNAM
5		Toomtarn Meenoi		TOOMTAM MEENOI
6		Kloy Raktinderm		KLOY RAKTINDERM
7				

# PROPER

(text)

abc efg → Abc Efg

ปรับตัวอักษรขึ้นต้นคำ (ภาษาอังกฤษ)  
จาก ตัวพิมพ์เล็ก ให้เป็น ตัวพิมพ์ใหญ่

และ

ปรับตัวอักษรที่ ไม่ใช่ คำขึ้นต้น  
จาก ตัวพิมพ์ใหญ่ ให้เป็น ตัวพิมพ์เล็ก

**=PROPER(B3)**

	A	B	C	D
1				
2		<b>Name Surname</b>		<b>หลังใช้ PROPER</b>
3		mr. som <b>C</b> hai saisamorn		Mr. Somchai Saisamorn
4		ms. khunchorn g <b>N</b> ODgnam		Ms. Khunchorn Gnodgnam
5		m <b>R</b> . toomtam meenoi		Mr. Toomtam Meenoi
6		mr. kloy r <b>A</b> ktinderm		Mr. Kloy Raktinderm
7				

# REPLACE

(old\_text, start\_num, num\_chars, new\_text)

ข้อความที่ต้องการเปลี่ยน

ลำดับของอักขระตัวแรกที่ต้องการ  
เปลี่ยน นับจากด้านซ้าย

จำนวนตัวอักษรที่ต้องการลบทิ้งไป โดยเริ่มนับ 1 จาก  
start\_num (ช่องว่าง 1 เค้านับเป็น 1 ตัวอักษร)

ข้อความใหม่ที่ต้องการแทนที่



ลบคำหรือตัวอักษรเก่าทิ้งไป  
แล้ววางข้อความใหม่เข้าไปแทน

1 2 3

6

"รัก"

=REPLACE(B5,20,3,"รัก")

=REPLACE(B4,11,3,"รัก")

=REPLACE(B3,6,3,"รัก")

1 2 3 4 5 7 8 9 11 12 13 14 15 16  
ผมแอบชอบคุณมานาน

	A	B	C	D
1				
2		ข้อความเดิม		แทนที่คำว่า "ชอบ" ด้วยคำว่า "รัก"
3		ผมแอบ <b>ชอบ</b> คุณมานาน		ผมแอบ <b>รัก</b> คุณมานาน
4		อยากบอกว่า <b>ชอบ</b> คุณเหลือเกิน		อยากบอกว่า <b>รัก</b> คุณเหลือเกิน
5		อยากรู้จังว่าเค้าจะ <b>ชอบ</b> เราบ้างหรือไม่		อยากรู้จังว่าเค้าจะ <b>รัก</b> เราบ้างหรือไม่
6				

# SUBSTITUTE



ค้นหาคำ/ตัวอักษร/สัญลักษณ์  
แล้วแทนที่ด้วยข้อความใหม่  
ที่ต้องการ  
(สามารถเลือกลำดับที่ต้องการ  
แทนที่ได้ด้วย)

(text, old\_text, new\_text, [instance\_num])

ข้อความที่ต้องการเปลี่ยน

คำหรือตัวอักษรที่ต้องการค้นหา

คำหรือตัวอักษรใหม่ที่ต้องการนำไปแทนที่

เลขลำดับของคำหรือตัวอักษรที่ต้องการเปลี่ยนเมื่อค้นเจอ  
(หากไม่ใส่เลขใดๆ จะหมายถึง ให้เปลี่ยนทุกคำที่ค้นเจอ)

**ค้นหาสัญลักษณ์และ  
เปลี่ยนเป็นคำว่าง**

**=SUBSTITUTE(B3,"^","")**

**ค้นหาและเปลี่ยนเฉพาะลำดับ  
ของคำที่ต้องการ**

**=SUBSTITUTE(B4,"ชอบ","รัก",3)**

**ค้นหาและเปลี่ยนทุกคำที่ค้นเจอ**

**=SUBSTITUTE(B3,"ชอบ","รัก")**

	A	B	C	D	E
1					
2		<b>ข้อความเดิม</b>		<b>ค้นหาและแทนที่ด้วย SUBSTITUTE</b>	
3		แอบ <b>ชอบ</b> คุณมานาน <b>ชอบ</b> คุณเหลือเกิน		แอบ <b>รัก</b> คุณมานาน <b>รัก</b> คุณเหลือเกิน	
4		จะบอก <b>ชอบ</b> <b>ชอบ</b> คุณทุกวัน		จะบอก <b>ชอบ</b> <b>ชอบ</b> คุณทุกวัน	
5		หวังว่าสักวันจะเปลี่ยนเป็นคำว่า <b>ชอบ</b>		หวังว่าสักวันจะเปลี่ยนเป็นคำว่า <b>รัก</b>	
6		^^^คิดถึงมากๆ ^^อยากอยู่ใกล้ๆ		คิดถึงมากๆ อยากอยู่ใกล้ๆ	



# VALUE (text)

แปลงค่าตัวเลขที่มี Format เป็น Text ให้  
กลับกลายเป็นตัวเลขจริงๆ

(มักจะพบบ่อยๆ ตอนที่แยกตัวเลขออกจากข้อความ  
โดยใช้ฟังก์ชันต่างๆ ที่กล่าวมาก่อนหน้า)

A B C → 1 2 3

**Text**

**Number**

**=VALUE(B3)**

	A	B	C
1			
2		<b>ก่อนใช้ VALUE</b>	<b>หลังใช้ VALUE</b>
3		\$2500	2500
4		15/11/19	43784
5		12:05 PM	0.503472222
6		12345	12345
7			

# CONCATENATE

(text1, [text2], ...)

คำหรือตัวอักษรที่ต้องการนำมาเชื่อมที่ 1

คำหรือตัวอักษรที่ต้องการนำมาเชื่อมที่ 2

คำหรือตัวอักษรที่ต้องการนำมาเชื่อมที่ n

ช่องว่าง 1 เคาะ

- มีหลักการเหมือนกับการใช้ **เครื่องหมาย &** มาเชื่อมข้อความ
- ใช้เชื่อมคำหรือตัวอักษรที่ต้องการ
- สามารถเชื่อมรายการได้สูงสุด 255 รายการ ไม่เกิน 8,192 อักขระ
- แต่อีกไม่นานจะถูกแทนที่ด้วยฟังก์ชัน **CONCAT** ซึ่งใช้งานง่ายกว่า แต่ยังไม่ให้ใช้ใน Excel2016 หรือต่ำกว่า



=CONCATENATE("แล้ว", " ",B6," ","ละ"," ",C6," ",D6," ","?")

=CONCATENATE(B5," ","ก็"," ",C5," ",D5," ","เหมือนกัน")

=CONCATENATE(B4," ",C4," ",D4)

=CONCATENATE(B3,C3,D3)

	A	B	C	D	E	F
1						
2		คำที่ 1	คำที่ 2	คำที่ 3		ใช้ CONCATENATE
3		ผม	รัก	คุณ		ผมรักคุณ
4		ผม	รัก	คุณ		ผม รัก คุณ
5		เขา	รัก	คุณ		เขา ก็ รัก คุณ เหมือนกัน
6		คุณ	รัก	ใคร		แล้ว คุณ ละ รัก ใคร ?
7						

# SUMPRODUCT +

## INDEX + MATCH

เลือกคอลัมน์ตาม Code  
แล้วหาผลรวม

**=SUMPRODUCT(C4:C7,**  
**INDEX(D4:F7,0,MATCH(H3,D3:F3,0)))**

เลข 0 ตรงนี้ หมายถึง ให้เลือกทุก  
Rows ในคอลัมน์ที่ต้องการ

เมื่อเลือกชื่อ Jeab

	A	B	C	D	E	F	G	H	I
1									
2				จำนวนที่ซื้อของแต่ละคน				ชื่อผู้ซื้อ	ค่าใช้จ่ายรวม
3		ชื่อสินค้า	ราคาสินค้าต่อหน่วย	Bob	Jeab	Beer		Jeab	245
4		สบู่	10	7	8	4			
5		ยาสีฟัน	25	3	2	6			
6		น้ำยาล้างจาน	20	6	2	5			
7		ผงซักฟอก	15	9	5	9			
8									
9									

$$=(10 \times 8) + (25 \times 2) + (20 \times 2) + (15 \times 5) = 245$$

78

# EDATE + YEARFRAC

คำนวณหาวันเกษียณอายุและจำนวนปีทำงานคงเหลือของพนักงาน

**=EDATE(D3,12\*60)**



**=YEARFRAC(TODAY(),E3)**

	A	B	C	D	E	F
1						
2		<b>ชื่อ</b>	<b>นามสกุล</b>	<b>วันเกิด</b>	<b>วันเกษียณอายุ (60 ปี)</b>	<b>จำนวนปีคงเหลือ ก่อนเกษียณ</b>
3		วันชัย	มีนา	15/05/1982	15/05/2042	21.1
4		วิรุทธ	สุดสมหวัง	12/07/1999	12/07/2059	38.3
5		ชนิดา	จันทร์ภา	09/08/1988	09/08/2048	27.4
6		พัชรินทร์	ธายาลัย	20/10/2000	20/10/2060	39.6
7		อุดมศักดิ์	จักรกรภาพร	02/02/1975	02/02/2035	13.8
8						



(logical\_test1, value\_if\_true1, ...)

เงื่อนไขที่ต้องการตรวจสอบ

คำตอบที่ต้องการ

ตรวจสอบหลาย  
เงื่อนไข

```
=IFS(F3<50,"F",
      F3<60,"D",
      F3<70,"C",
      F3<80,"B",
      F3<=100,"A")
```

	A	B	C	D	E	F	G
1							
2		คะแนน	Grade		ชื่อ	คะแนน	Grade
3		0	F		วินัย	59	D
4		50	D		มานี	75	B
5		60	C		อาชา	66	C
6		70	B		อินทิรา	45	F
7		80	A		ชูใจ	90	A



# TEXTJOIN

(delimiter, ignore\_empty, text1, ...)

สัญลักษณ์ที่ต้องการใช้คั่นคำ

ไม่รวมเซลล์  
ที่เป็นช่องว่างใช่หรือไม่

เซลล์หรือช่วงเซลล์ที่  
ต้องการเชื่อมคำ

**=TEXTJOIN(" ", TRUE, B3:B7)**

	A	B	C	D
1				
2		Words		เชื่อมคำแบบมีช่องว่างคั่น
3		I		I Love You Very Much
4		Love		
5		You		
6		Very		
7		Much		
8				

# LOOKUP ข้อมูล **แบบนอน**

ด้วย **INDEX+MATCH**

```
=INDEX($C$3:$N$3,  
MATCH(C8,$C$2:$N$2,0))
```

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1															
2		Type/Month	Jan-19	Feb-19	Mar-19	Apr-19	May-19	Jun-19	Jul-19	Aug-19	Sep-19	Oct-19	Nov-19	Dec-19	Total
3		Units Sold	106	238	26	102	221	74	42	21	90	41	13	263	1,237
4															
5															
6		2 Months Compare													
7															
8		Type/Month	Jun-19	Dec-19											
9		Units Sold	74	263											
10															
11															
12															
13															







# XLOOKUP

ค้นหาทางเดียวแนวตั้งแบบ **VLOOKUP** หรือ INDEX+MATCH

**=XLOOKUP(F3,B3:B10,D3:D10)**



	A	B	C	D	E	F	G
1							
2		<b>ชื่อเมนูอาหาร</b>	<b>หน่วย</b>	<b>ปริมาณแคลอรี (Kcal)</b>		<b>ชื่อเมนูอาหาร</b>	<b>ปริมาณแคลอรี (Kcal)</b>
3		กระเพาะปลา	ชาม	150		กล้วยคลูกมะพร้าว	300
4		กระเพาะปลาตุ๋นน้ำแดง	ชาม	225			
5		กล้วยไข่	ลูก	40			
6		กล้วยคลูกมะพร้าว		300			
7		กล้วยฉาบ	ชิ้น	29			
8		กล้วยตาก	ผล	30			
9		กล้วยทอด (กล้วยแขก)	ชิ้น	50			
10		กล้วยน้ำว้า	ลูก	36			
11							

Diagram illustrating the XLOOKUP function. A red arrow points from the glasses icon to the search range (B3:B10). A red arrow points from the value 300 in cell D6 to the search range (B3:B10). A yellow dashed arrow points from the value 300 in cell D6 to the result cell G3. A black line points from the formula bar to the result cell G3.

# XLOOKUP

ค้นหาทางเดียวแบบ **HLOOKUP** หรือ INDEX+MATCH

**=XLOOKUP(C\$10,\$C\$2:\$O\$2,\$C3:\$O3)**



	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1															
2			Jan-19	Feb-19	Mar-19	Apr-19	May-19	Jun-19	Jul-19	Aug-19	Sep-19	Oct-19	Nov-19	Dec-19	Total
3		Units Sold	106	238	26	102	221	74	42	21	90	41	13	263	1,237
4		Expense	16,100	31,850	3,200	14,880	32,240	10,350	5,850	3,150	11,960	4,950	1,650	35,700	171,880
5		Cost Per Unit	152	134	123	146	146	140	139	150	133	121	127	136	139

## 2 Months Compare

Type/Month	Jun-19	Dec-19
Units Sold	74	263
Expense	10,350	35,700
Cost Per Unit	140	136



87

# XLOOKUP

**ข้อ Error** โดย

ไม่จำเป็นต้องใช้ IFERROR หรือ IFNA

**=XLOOKUP(D3,B3:B7,B3:B7,"ค้นหาไม่พบ")**

	A	B	C	D	E
1					
2		<b>ชื่อ-นามสกุล</b>		<b>คำค้น</b>	<b>ผลลัพธ์</b>
3		บุญเต็ม เพิ่มสัจจา		นพพร ใจดี	ค้นหาไม่พบ
4		จินตนา ลาสำ			
5		สุรเชษฐ์ เยี่ยมสุขสันต์			
6		กุลนันท์ จันทรา			
7		ศาสตรา พาเจริญ			
8					
9					

# XLOOKUP

ทำงานร่วมกับ Wildcard

ค้นหาได้แม้คีย์คำค้นแค่บางส่วน

**=XLOOKUP("\*"&D3&"\*",B3:B7,B3:B7,"ค้นหาไม่พบ",2)**

	A	B	C	D	E
1					
2		<b>ชื่อ-นามสกุล</b>		<b>คำค้น</b>	<b>ผลลัพธ์</b>
3		บุญเต็ม เพิ่มสัจจา		สุรเช	สุรเชษฐ์ เยี่ยมสุขสันต์
4		จินตนา ลาสำ			
5		สุรเชษฐ์ เยี่ยมสุขสันต์			
6		กุลนันท์ จันทรา			
7		ศาสตรา พาเจริญ			
8					
9					

Diagram illustrating the XLOOKUP function with wildcards. The search key in cell D3 is "สุรเช" (highlighted in a dashed box). The function searches for this key in the range B3:B7. The first match is found in cell B5, which contains "สุรเชษฐ์ เยี่ยมสุขสันต์" (highlighted in a dashed box). An arrow points from the search key to the first match. Another arrow points from the result cell E3 to the first match cell B5. A vertical line points from the result cell E3 to the formula bar above.



# ค้นหารายการทั้งหมด

แม้คีย์คำค้นแค่บางส่วน  
ก็ค้นหาได้

**=ISNUMBER(SEARCH(\$F\$4,\$B\$5:\$B\$17))**

**=IFERROR(  
INDEX(\$B\$5:\$B\$17,  
SMALL(IF(ISNUMBER(SEARCH(\$F\$4,\$B\$5:\$B\$17)),  
ROW(\$B\$5:\$B\$17)-4),H7)),")**

**=IFERROR(VLOOKUP(I7,\$B\$5:\$C\$17,2,0),")**

A	B	C	D	E	F	G	H	I	J	K
1										
2	<b>ฐานข้อมูล</b>		<b>ระบบค้นหาทะเบียนรถจากฐานข้อมูล</b>							
3										
4	<b>ทะเบียนรถ</b>	<b>จังหวัด</b>			พิมพ์ตัวอักษรหรือตัวเลขเพื่อค้นหา	5		มีทั้งหมด	5	ค้น
5	กท1234	กรุงเทพ								
6	จม9997	กรุงเทพ								
7	ทก1237	กรุงเทพ								
8	ขฎ7963	ลพบุรี								
9	7กร182	เชียงใหม่								
10	กต1855	ลำปาง								
11	5ขม6079	เชียงใหม่								
12	พฟ5678	สุพรรณบุรี								
13	ดร389	กรุงเทพ								
14	3กก9999	กรุงเทพ								
15	ปก2356	ลำปาง								
16	พร7	กรุงเทพ								
17	9กต566	กรุงเทพ								
18										

ลำดับที่	ทะเบียนรถ	จังหวัด
1	กต1855	ลำปาง
2	5ขม6079	เชียงใหม่
3	พฟ5678	สุพรรณบุรี
4	ปก2356	ลำปาง
5	9กต566	กรุงเทพ

# LEN + SUBSTITUTE

**นับคำที่ต้องการ** ในข้อความที่อยู่ในเซลล์เดียว

$$=(LEN(B3)-LEN(SUBSTITUTE(B3,C3,"")))/LEN(C3)$$

	A	B	C	D
1				
2		<b>ข้อความ</b>	<b>คำที่ต้องการนับ</b>	<b>จำนวน</b>
3		ผมรักคุณมาก รักตั้งแต่แรกพบ รักสุดหัวใจ ยิ่งเวลาผ่านไปนานเท่าไร ยิ่งรักมากขึ้น เท่านั้น ผมสัญญาจะรักษาความรักของเราให้คงอยู่ตลอดไป ไม่ว่าจะนานแสนนานแค่ไหน ก็จะรักคุณตลอดไปไม่เปลี่ยนแปลง	รัก	7
4		อยู่ที่เรียนรู้ อยู่ที่ยอมรับมัน ตามความคิดสติเราให้ทัน อยู่กับสิ่งที่มี ไม่ใช่สิ่งที่ฝัน	อยู่	3
5		แต่อย่ามาจี้จี้ให้มันมากไป มันจะเป็นบ้า เรื่องเล็กจะกลายเป็นเรื่องใหญ่ หากเปลี่ยนเป็นจี้จี้จะดีกว่าไหม ฉันว่าเข้าท่าแต่ถ้าจี้จี้ฉันก็ต้องไป	จี้จี้	2
6				

มีคำ ก็นับได้  
เลยครับ



# FREQUENCY

**=FREQUENCY(C5:C104,H5:H12)**

ข้อมูลอายุของพนักงานจำนวน 100 คน

คนที่	อายุ
1	43
2	41
3	30
4	34
5	57
6	47
7	48
8	37
9	33
94	40
95	57
97	45
98	35
99	50
100	55

กลุ่มที่	ช่วงอายุ	MAX	จำนวน
1	21 - 25	25	2
2	26 - 30	30	9
3	31 - 35	35	19
4	36 - 40	40	20
5	41 - 45	45	17
6	46 - 50	50	14
7	51 - 55	55	11
8	56 - 60	60	8

สร้างตารางแจกแจงความถี่ข้อมูล

เสร็จแล้วกด 3 ปุ่มนี้ครับ

**Ctrl + Shift + Enter**



# สูตรหาค่าสูงสุด

แบบมีเงื่อนไข

MAXIFS

All Version

2019, 2021, Microsoft 365

**{=MAX(IF(C3:C11=F3,D3:D11))}**

**=MAXIFS(D3:D11,C3:C11,F3)**

Ctrl + Shift + Enter

Enter

\*\* ตอนเขียนสูตรไม่ต้องพิมพ์ { } นะครับเดี่ยวมันจะปรากฏขึ้นมาเองตอนกด 3 ปุ่มด้านบนนี้

	A	B	C	D	E	F	G
1							
2		<b>Date</b>	<b>Product</b>	<b>Total Revenue</b>		<b>Product</b>	<b>Max Revenue</b>
3		Jan-21	โพสอิท	33,250		สมุดโน้ต	21,000
4		Jan-21	ดินสอ	4,000			
5		Jan-21	สมุดโน้ต	2,880	✘		
6		Feb-21	โพสอิท	2,700			
7		Feb-21	ดินสอ	4,480			
8		Feb-21	สมุดโน้ต	21,000	✔	<b>MAX</b>	
9		Mar-21	โพสอิท	3,750			
10		Mar-21	ดินสอ	7,200			
11		Mar-21	สมุดโน้ต	8,350	✘		
12							



# สูตรหาค่าต่ำสุด

ไม่สนใจ ค่าติดลบ และ ค่า 0

\*\* ตอนเขียนสูตรไม่ต้องพิมพ์ { } นะครับ  
เดี๋ยวมันจะปรากฏขึ้นมาเองตอนกด 3 ปุ่ม  
ด้านล่างนี้

All Version

Excel 2019 Up

Microsoft 365

**{**=MIN(IF(C3:C9>0,  
C3:C9))**}**

=**MINIFS**(C3:C9,  
C3:C9,">0")

=INDEX(C3:C9,  
**XMATCH**(0.00001,  
C3:C9,1))

Ctrl + Shift + Enter


Enter

Enter

	A	B	C	D	E
1					
2		<b>Time</b>	<b>Value</b>		<b>Min Value</b>
3		08:00	-2.09		1.99
4		08:10	4.99		
5		08:20	2.57		
6		08:30	-0.78		
7		08:40	1.99	<b>✓ MIN</b>	
8		08:50	0.00		
9		09:00	6.80		
10					

"ไม่สนใจค่าลบ  
และค่า 0"





# สูตร ตัดคำซ้ำ

## UNIQUE

All Version

2021, Microsoft 365

```
{=INDEX($B$3:$B$10,
MATCH(0,
COUNTIF($D$2:D2,
$B$3:$B$10),0))}
```

=UNIQUE(B3:B10)

Ctrl + Shift + Enter

Enter

\*\* ตอนเขียนสูตรไม่ต้องพิมพ์  
{ } นะครับเดี๋ยวมันจะปรากฏ  
ขึ้นมาเองตอนกด 3 ปุ่ม  
ด้านบนนี้

	A	B	C	D	E
1					
2		สินค้า		Unique	
3		น้ำอัดลม	✓	น้ำอัดลม	
4		น้ำอัดลม		ปลากระป๋อง	
5		ปลากระป๋อง	✓	นมสด	
6		ปลากระป๋อง		ขนมปังสำเร็จรูป	
7		นมสด	✓		
8		ขนมปังสำเร็จรูป	✓		
9		น้ำอัดลม			
10		ขนมปังสำเร็จรูป			


# จัดลำดับข้อมูล จากแนวตั้งเป็นแนวนอน

## ตามจำนวนแถวและคอลัมน์ที่ต้องการ

**=INDEX(B3:B14,SEQUENCE(3,4))**

	A	B	C	D	E	F	G	H
1								
2		<b>Product</b>						
3		A		A	B	C	D	3
4		B		E	F	G	H	
5		C		I	J	K	L	
6		D						
7		E						
8		F						
9		G						
10		H						
11		I						
12		J						
13		K						
14		L						
15								

4



**พิมพ์สูตรที่เซลล์ D3 แล้ว  
Enter ได้เลยครับ**

# หาผลรวมระหว่าง 2 วัน

ง่ายๆ ด้วย

# SUMIFS

**=SUMIFS(D3:D12,**  
**B3:B12, ">="&G2,**  
**B3:B12, "<="&G3)**



	A	B	C	D	E	F	G
1							
2		<b>Date</b>	<b>Product</b>	<b>Units</b>		<b>Start Date</b>	<b>01/02/2021</b>
3		06/01/2021	แฟ้ม	95		<b>End Date</b>	<b>30/04/2021</b>
4		23/01/2021	ไฮไลท์	50			
5		09/02/2021	ไฮไลท์	36		<b>Sum Units</b>	<b>254</b>
6		26/02/2021	สมุดโน้ต	27			
7		15/03/2021	ไฮไลท์	56			
8		01/04/2021	แฟ้ม	60			
9		18/04/2021	ดินสอ	75			
10		05/05/2021	ไฮไลท์	90			
11		22/05/2021	สมุดโน้ต	32			
12		08/06/2021	แฟ้ม	60			
13							

# Get fiscal quarter from Date

= "Q"&CHOOSE(MONTH(B3),1,1,1,2,2,2,3,3,3,4,4,4)

= "Q"&CHOOSE(MONTH(B3),4,4,4,1,1,1,2,2,2,3,3,3)

= "Q"&CHOOSE(MONTH(B3),3,3,3,4,4,4,1,1,1,2,2,2)

= "Q"&CHOOSE(MONTH(B3),2,2,2,3,3,3,4,4,4,1,1,1)

	A	B	C	D	E	F
1						
2		Date	Jan	Apr	Jul	Oct
3		10 Jan 21	Q1	Q4	Q3	Q2
4		15 Feb 21	Q1	Q4	Q3	Q2
5		30 Mar 21	Q1	Q4	Q3	Q2
6		08 Apr 21	Q2	Q1	Q4	Q3
7		26 May 21	Q2	Q1	Q4	Q3
8		06 Jun 21	Q2	Q1	Q4	Q3
9		12 Jul 21	Q3	Q2	Q1	Q4
10		23 Aug 21	Q3	Q2	Q1	Q4
11		24 Sep 21	Q3	Q2	Q1	Q4
12		27 Oct 21	Q4	Q3	Q2	Q1
13		02 Nov 21	Q4	Q3	Q2	Q1
14		11 Dec 21	Q4	Q3	Q2	Q1
15						

เริ่มต้น Q1 เดือนไหน  
เลือกสูตรได้เลยครับ



Start Jan

Start Apr

Start Jul

Start Oct



วิศวกรสร้างภาพ